

### 3.1 プログラミング fn

ver. 0.2

第3節では GAUSS プログラミングの基礎とその作法についてくわしく見ていきます。GAUSS では他の統計計量のパッケージソフトと違って、いくつかの procedure を作ってそれをつなぎ合わせることによって、プログラムを組み立てていく手法をとるのが正統的なやり方です。まずは、いわば procedure のミニチュア版とも言えるファンクション fn を用いた1行で表される関数の定義について見てみます。

関数定義 fn は、非常に便利な機能で、主に procedure の作法にのっとってプログラムを前方参照させることなしに、手軽にプログラムの流れの中に関数形を定義する際に使われます。たとえば、 $f(x)=3x^3-4x^2+2x-7$  という関数をあらかじめ定義しておいて、 $f(9)$ を求めるやり方は次のようになります。

プログラム

```
new; cls;
fn f(x)=3*x^3-4*x^2+2*x-7;
y=f(9);
print /rz y;
```

画面表示

1874

上のように、fnは関数を定義する際に用いて、プログラムでそれ以降、普通の代数の計算と同じように、関数に値を代入して計算することができます。なお、関数名は何であってもかまいませんし、変数名も何であってもかまいません。上の例は、関数名を f、変数を x とした場合の1変数の例です。関数f(x)をfnで定義した後、その f に9を代入して計算したものを y とおいて、それを右寄せ小数点以降の連続する0を省略するフォーマットで画面表示したものです。

もちろん、関数名を f 以外のものにしたり、2変数、3変数にすることも可能です。

プログラム

```
new; cls;
fn f1(x,y)=x^(2/3)*y^(2/3);
a=f1(4,7);
print "      f1(4,7)=" a;
fn f2(x,y,z)=sin(x)+cos(y)+tan(z);
b=f2(0.5,0.5,0.5);
print "f2(0.5,0.5,0.5)=" b;
```

画面表示

```
      f1(4,7)=      9.2208726
f2(0.5,0.5,0.5)=    1.9033106
```

画面表示を見やすくするためにprint文に引用符でくるんで文字を表示させていますので、すこし複雑に見えますが、内容としては、第1番目のプログラムと同様に、fnで関数を定義して、その関数に数値を代入して、それを別の変数とおいた上で、画面表示させています。2変数の関数の場合には関数名のあとの括弧のなかにカンマをつけて任意の変数を入れて、その2変数を使ってイコール以降に関数形を書きます。3変数の場合も同様で、括弧のなかにカンマをつけて任意の変数を入れて、その3変数を使って、イコール以降に関数形を書きます。その利用の仕方は、1変数の場合と全く同じように、関数の中に、実際の数値を複数個入れて計算させます。

上の2つの例のように定義される関数がスケーラ変数であることは、GAUSSプログラミングではまれで、むしろ、行列、少なくとも列ベクトルとして使われていることが大多数です。関数定義fnでは各変数のディメンションを明示的に書くことはできませんが、その関数にかけ合わせる変数のディメンションと代入される数値によってGAUSSは、その関数がスケーラなのか列ベクトル（または行列）なのかを決定します。まずは、代入する数値によって、同じ定義された関数が、スケーラにも列ベクトルにも（あるいは行ベクトルにも）行列にもなり得る例を示します。

#### プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
a={1,2,3}; b={1 2 3}; c={1 2 3, 4 5 6, 7 8 9};
print /rz f(7); print /rz f(a); print; print /rz f(b); print /rz f(c);
```

#### 画面表示

```
931

7
36
99

7          36          99

7          36          99
208        375        612
931        1344       1863
```

上の例は、関数定義fnで $f(x)=2x^3+5x^2$ と定義しておいて、右寄せ連続する小数点以下の0を省略する/rzフォーマットで $f(x)$ にスケーラ7、列ベクトルa、行ベクトルb、そして3×3の行列cを代入した結果を示しています。なお、GAUSSでは、行ベクトルを表示させる場合には、標準的な行列表示のように1行送りをしませんが、print;で1行挿入してい

ます。上の画面出力結果のように、同じ関数でも、代入する数値の型にあわせて代入結果はかわってきます。

もちろん、上のように代入される数値によって関数のディメンションが決まるのが優先されるのですが、関数内の変数にかかる数がスケーラーかベクトルによって、行列の `comformable` の計算上の制約から関数がスケーラーなのかベクトル（あるいは行列）なのかも決定されます。例えば、

```
a1=7; b1=9;  
fn f1(x)=a1*x*b1;
```

というふうに設定すると、関数 `f 1` は一般的な行列の関数として定義されます。スケーラーも GAUSS では  $1 \times 1$  の特殊な行列として扱われます。もちろん、`x` のところに列ベクトルを代入すれば、関数 `f 1` は同じ関数ばかりの集まりの列ベクトルになることは、以前の例からも明らかです。`x` にスケーラーを代入すれば、結果はスケーラーで返されます。しかしながら、次の場合には、必ず、関数は行列として定義されます。

```
a2={9 8 2}; b2={3 9 1, 2 4 8, 5 1 2};  
fn f2(x)=a2*x*b2;
```

`x` に右からかかる `a2` は  $1 \times 3$ 、左からかかる `b2` は  $3 \times 3$  ですから、`x` は  $3 \times 3$  であるはずで、`x` に  $3 \times 3$  の数値を与えてやると、`f 2` の返す結果は  $1 \times 3$  の行ベクトルになるはずで、このように、GAUSS では、データを読み取る際以外には、変数の型やディメンションの宣言をしないのが普通で、プログラムする際には、変数のディメンションを考えながらアルゴリズムをユーザーサイドで組み立てなければなりません。プログラムの読み手も、もとのデータのディメンションや各行列変数の `comformable` な関係をたどって、どういうディメンションの計算が行なわれているのかを理解する必要があります。これは、関数定義 `fn` や `proc` に限ったことではなくて、GAUSS プログラミング全般に言えます。

**ポイント** `fn 関数名(x,y,z,...)=関数式;` `fn` は「1行で」表現できる関数を定義する

さて、この `fn` は非常に便利なのですが、一体どういう場合に使えばいいのでしょうか。GAUSS のマニュアル関係には、微積関係の組込み関数を用いる場合、`proc` の手法で関数形を定義して、その `proc` をプログラムの一番おしまいにおいて、それらの微積の組込み関数の中で前方参照させるように解説されています。また、多くの公開されているスクリプトにもその方法が踏襲されています。しかしながら、関数定義 `fn` の本当の使い道は、エラーや分岐を考えなくてもよい、1行で定義される関数、すなわち微積の組込み関数で使われるような関数形だけがほしいような場合を念頭においてもともと作られた機能であると言えます。マニュアルに記述が全くないために、以下のような使われ方をすることはまれですが、実際にはそう使われるべきなのです。以下、微積関係の組込み関数の使い方を関数定義 `fn` との関連で説明していきます。

まずは、ある座標上の数値で評価された gradient を求める方法です。まず、関数 f(x) を定義したあとに gradp という組込み関数で、その関数 f を呼び出して、評価されるべき値を代入します。ここでは、関数は単一変数の列ベクトルである必要があります。

プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
x0=1;
y=gradp(&f,x0);
print y;
```

画面表示

16.000000

上の例は、関数  $f(x)=2x^3+5x^2$  と fn で定義したうえで、 $x_0=1$  で評価される gradient を求めています。すなわち、 $f(x)=6x^2+10$  に 1 を代入した値と同じものを求めていることになります。gradient を求める組込み関数 gradp は、その第 1 要素に定義された関数名に & のマークをつけておき、第 2 要素には評価すべき場所の座標を入れます。この場合、スケーラー 1 が  $x_0$  とおかれたうえで代入されています。ただし、原理的に GAUSS 内部では、Exact な微分した式に値を代入しているのではなくて、微分の定義式

$$y = f'(x) = \frac{f(x+h) - f(x)}{h}, \quad \text{as } h \rightarrow 0$$

に基づいた近似値を出しています。すなわち、

プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
x0=1;
h=1e-8;
y=(f(x0+h)-f(x0))/h;
print y;
```

というふうに、h のところに例えば 1e-8 というような十分に小さな値を入れることによって、線形近似を行なっています。これにより、この場合は誤差なく答えは 16 と出ます。

今度は、評価する値が 1 値ではなくて列ベクトルであると次のようになります。

プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
x0={1,2,3};
y=gradp(&f,x0);
print y;
```

画面表示

16.000000	0.00000000	0.00000000
0.00000000	44.000000	0.00000000
0.00000000	0.00000000	84.000001

この場合には、以前と同じ関数 $f(x)$ を使って、評価するポイントを列ベクトル $\{1,2,3\}$ にして gradient を求めています。関数 $f(x)$ に  $3 \times 1$  の値を入れて評価するのですから、関数 $f(x)$ も  $3 \times 1$  の列ベクトルになります。以前と同様、組込み関数 gradp では、ファイル名に & のマークをつけて呼び出し、評価すべき列ベクトル  $x_0$  を代入しています。なお、評価する値は、必ず列ベクトルまたはスケーラーである必要があります。カンマなしの行ベクトル、または 2 列以上の行列は不可です。画面結果でわかるように、diagonal は列ベクトル  $f(x)$  の 1 行目から 3 行目まで同じ形の関数であるため、関数を微分したそれぞれに 1、2、3 を順番に代入した形になっています。ただし、off diagonal はすべて 0 になっています。ちょうど変数  $x$  の第 1 要素が  $x$ 、第 2 要素が  $y$ 、第 3 要素が  $z$  という具合になっていて、偏微分の評価値が 0 になっているのです。上の 84.000001 は、明らかに、本来は 84 ですが、gradp の線形近似のアルゴリズムにより計算の小数以下が若干上方にぶれる傾向があります。したがって、実際に GAUSS の微積で出た結果には注意が必要ですし、小数以下 3 ~ 4 桁より下の値には信頼性がありません。ある桁数を決めて、ラウンドするか、フォーマットで小数の下の方の値は表示されないようにする必要があります。

次に、同じ関数  $f(x)$  で、Hessian を求めてみましょう。やり方は gradp とほぼ同じです。

プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
x0={1,2,3};
y=hessp(&f,x0);
print y;
```

画面表示

21.999611	-0.00038754908	-0.00025836606
-0.00038754908	34.000263	0.00000000
-0.00025836606	0.00000000	46.000354

上の結果は明らかに誤りです。組込み関数 gradp と同様、小数点以下 3 ~ 4 桁より小さい桁の精度は保証されていません。上の diagonal は整数であるはずですから、右寄せ十進法表示、16 文字スペース分確保、小数点以下 3 桁のフォーマットである、

```
format /rd 16,3;
```

という行を print 文の前に書き入れれば、おおよそのところの結果がでます。

画面表示

22.000	0.000	0.000
--------	-------	-------

0.000	34.000	0.000
0.000	0.000	46.000

上の結果のoff diagonalはすべて0、diagonalは整数になっています。このようにGAUSSでは、微分にかかわる計算には、最大級の慎重さを要します。必要であれば、数学系のソフトでgradientやHessianを別途求めてから、GAUSS上に代入するのも手かもしれません。組込み関数hesspでは、gradpと同様に、&マークつきの関数名を呼び出して、列ベクトルの評価座標を与えます。上の例で言えば、x0に相当する部分は、必ず列ベクトル（またはスケーラーであることが必要です。結果は、対称行列になります。

ポイント gradp(&関数名,評価列ベクトル)                      gradientの行列を返す  
 ポイント hessp(&関数名,評価列ベクトル)                      Hessianの行列を返す  
 gradp、hesspともに小数点以下3～4桁よりも小さい桁数は保証されず  
 関数名はfnまたはprocであらかじめ定義する必要があります

より一般的なoff diagonalも非零になる例は、以下のようなプログラムになります。結果が対称行列になっていることがよくわかると思います。

プログラム

```
new; cls;
a={1 2 3};
fn f(x)=exp(a*x);
x0={3,2,1};
y=hessp(&f,x0);
format /rd 16,3; print y;
```

画面表示

22026.854	44053.620	66080.562
44053.620	88107.629	132160.860
66080.562	132160.860	198240.993

計量の計算では、行列のままで計算するので用が足りませんが、経済数学として利用するには、 $x[1]$ や $x[2]$ といったディメンションの簡略形を $x_1$ や $x_2$ の代わりに使うことをよくします。この方法を用いたやり方で $f(x_1, x_2) = 3x_1^3 x_2^2$ のgradientとHessianを座標(1,2)について求めてみましょう。以下がプログラムです。小数点第3位よりも小さい桁は有効ではありませんので、format文で見た目上まるめます。

プログラム

```
new; cls;
fn f(x)=3*x[1]^3*x[2]^2;
x0={1,2};
```

```
format /rd 16,3;
print gradp(&f,x0);
print hessp(&f,x0);
```

画面表示

```
36.000      12.000

72.000      36.000
36.0        6.000
```

上の最初の 1 行 2 列の結果は、それぞれ gradient である  $f_1(1,2)$ 、 $f_2(1,2)$  の計算結果です。その次の  $2 \times 2$  の行列の結果は、それぞれ Hessian である  $f_{11}(1,2)$ 、 $f_{12}(1,2)$ 、そして次の行に移って  $f_{21}(1,2)$ 、 $f_{22}(1,2)$  の計算結果です。

さて今度は積分です。fn で定義された関数の定積分を求める方法を説明します。1 つ前の  $f(x)$  をそのまま使って、0 から 1 までの区間を定積分してやりましょう。

プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
x1={1, 0};
format /ro 24,16;
print "----- | 123456789 | 12345--";
y=intsimp(&f,x1,1e-11); print "1e-11   :" y;
y=intsimp(&f,x1,1e-10); print "1e-10   :" y;
y=intsimp(&f,x1,1e-9); print "1e-9     :" y;
y=intsimp(&f,x1,1e-8); print "1e-8     :" y;
y=intsimp(&f,x1,1e-7); print "1e-7     :" y;
y=intsimp(&f,x1,1e-6); print "1e-6     :" y;
y=intsimp(&f,x1,1e-5); print "1e-5     :" y;
z=intquad1(&f,x1); print "Legendre " z;
```

画面表示

```
----- | 123456789 | 12345--
1e-11   :      2.1666666666660309
1e-10   :      2.1666666666641235
1e-9    :      2.1666666666259766
1e-8    :      2.166666660156250
1e-7    :      2.166666640625000
1e-6    :      2.166666640625000
1e-5    :      2.166666640625000
```

Legendre 2.1666666666666666

上の例では、 $f(x)$  についての 0 から 1 までの区間の定積分を 2 つの方法で計算させています。1 つは、第 3 要素に有効桁数を指定してシンプソン法で計算する組込み関数 `intsimp`。もう 1 つは、ガウス＝ルジャンドル求積法で計算する組込み関数 `intquad1`。どちらの関数も、その第 1 要素に関数名に `&` のマークをつけて呼び出し、第 2 要素に  $2 \times 1$  の区間を表す列ベクトルを入れます。0 から 1 までを区間として指定するのには、 $\{1, 0\}$  というふうに、インテグラルの上にくる区間の方をまず書いて、カンマで区切って、インテグラルの下にくる方を書きます。逆にしてしまうと、当然のことながら、マイナスの数になってしまいます。一部の書物には、`intsimp` の方がより精度の高い計算をするという記述がありますが、上の例でみるように、手計算の定積分の結果は  $13/6$  ですから、ガウス＝ルジャンドル求積法の方が正確になるケースもあり得るわけです。ケースバイケースで使い分けるとよいのではないのでしょうか。なお、シンプソン法は、有効桁数を標準で指定できますが、区間を表す第 2 要素は  $2 \times 1$  の 1 組分の計算しかできません。しかしながら、ガウス＝ルジャンドル求積法の方は、2 組以上の区間の定積分が可能です。グローバル変数を使えば、有効桁数の変更も可能です。以下が、その例です。

#### プログラム

```
new; cls;
fn f(x)=2*x^3+5*x^2;
x1={2 2.5 3, 0 0 0};
format /ro 16,12;
z=intquad1(&f,x1);
print z;
```

#### 画面表示

```
21.3333333333
45.5729166667
85.5000000000
```

上の例は、定積分の区間が、0 から 1 ではなくて、0 から 2、0 から 2.5、0 から 3 の 3 組についてまとめて計算した結果です。このように  $2 \times 1$  の列ベクトルの区間の組を水平方向に作っていくことによって、まとめて定積分できます。なお、この機能はシンプソン法にはありません。

上の組込み関数 `intquad1` の 1 は普通の 1 重の定積分のことで、2 重積分の場合には 2 が、3 重積分の場合には 3 が、1 の代わりにつきます。当然のことながら、`fn` で定義する関数は、2 重積分の場合には 2 変数が、3 重積分の場合には 3 変数が、それぞれ必要になってきます。区間の方も、2 重積分の方は 2 組、3 重積分の方は 3 組、それぞれ必要になってきます。以下がその簡単な場合についてです。まず、下の二重定積分を計算してみます。



$$\int_0^1 \int_0^1 xy dx dy$$

#### プログラム

```
new; cls;
fn f(x,y)=x.*y;
x1={1,0}; y1={1,0};
a=intquad2(&f,x1,y1);
print a;
```

#### 画面結果

```
0.25000000
```

二重定積分の組込み関数intquad2を用いるのには、変数を2つもつ関数を定義した上で、積分する区間も2変数分必要になります。関数の名前は何でもいいのですが、仮にfとして、そこでの2変数をxとyとします。関数を定義するためにfnを使って、f(x,y)=数式;とすれば関数は定義されます。通常の数学とここまでは同じです。ただし、intquad系の組込み関数は、積分の区間の複数の組を1度に計算するようなアルゴリズムで成り立っているため、数式部分は、同じ式の集まりの列ベクトルであることを許すような式でなければなりません。そのため、スケーラ対スケーラを念頭に置いたx\*yではなくて、列ベクトルの要素対要素を念頭に置いたx.\*yになっています。掛け算同様、割り算の場合も./の要素対要素の演算を用いる必要があります。そして、xの区間を0から1までという意味で、{1,0}と指定して(視覚的に1が上になっています。これを逆転させると積分の値はマイナスになります) yの区間を0から1まで{1,0}と指定してから、intquad2で関数fを&のマークをつけて呼び出します。その第2要素にはxの区間の入った変数、第3要素にはyの区間の入った変数を入れます。

2組以上の定積分の組をするのも、1重積分の場合と同じです。区間の変数のところを行列代入します。1組目がxが0から1で、yも0から1、2組目がxが0から2、yが0から3とすれば、以下のような指定をします。

#### プログラム

```
new; cls;
fn f(x,y)=x.*y;
x1=[1 2,0 0]; y1=[1 3,0 0];
a=intquad2(&f,x1,y1);
print a;
```

#### 画面表示

```
0.25000000
```

```
9.00000000
```

次に、三重定積分も同様に計算できることを示します。

## プログラム

```
new; cls;
fn f(x,y,z)=x.*y.*z;
x1={1,0}; y1={1,0}; z1={1,0};
a=intquad3(&f,x1,y1,z1);
print a;
```

## 画面表示

0.12500000

上のように、3変数の関数 $f(x,y,z)$ をfnで定義して、 $x$ 、 $y$ 、 $z$ の3区間を決定し、組み関数intquad3で関数を&つきで呼び出して、3変数の区間をそれぞれ順に代入します。定義される関数の中で、列ベクトルを念頭に置いた要素対要素のドットのついた演算をするのは以前の例と同じです。

ポイント	intsimp( &関数名, $x$ の区間, 有効計算精度桁数)	シンプソン法による積分
ポイント	intquad1( &関数名, $x$ の区間)	一重積分
ポイント	intquad2( &関数名, $x$ の区間, $y$ の区間)	二重積分
ポイント	intquad3( &関数名, $x$ の区間, $y$ の区間, $z$ の区間)	三重積分

定積分には、このほかに、二重積分の場合に、 $y$ の区間に $x$ の関数を入れる方法に対する組み関数intgrat2、三重積分の場合に、 $y$ の区間に $x$ の関数を、 $z$ の区間に $x$ と $y$ からなる関数を入れる組み関数intgrat3が用意されています。まずは、二重積分から

## プログラム

```
new; cls;
fn f(x,y)=x.*y;
fn g1(x)=x^2+x./4+3;
fn g2(x)=3;
x1={1,0}; g=&g1 | &g2;
a=intgrat2(&f,x1,g);
print a;
```

## 画面表示

1.1411458

上の例は、以前と同じ $f(x,y) = xy$ を二重積分するものですが、 $x$ の区間は0から1、しかしながら、 $y$ に相当する区間は、3から $g(x)=x^2+x/4+3$ まで $x$ の区間にしたがって、積分するものです。したがって、関数 $g$ には、この $g1(x)=x^2+x/4+3$ と定数3を垂直方向に|のマークでマージさせた $2 \times 1$ のベクトルが入ります。ただし、これらは数値ではなく、定数3も含めて関数ですので、 $g1$ 、 $g2$ を指すものとして&をつけて、&g1と&g2を垂直方向にマ-

ジさせたものを  $g$  と置きます。組込み関数 `intgrat2` には、その第 1 要素に関数  $f$  の指すもの、つまり  $\&f$  が入り、第 2 要素には  $x$  の区間を表す変数、第 3 要素には  $y$  の区間を表す変数が入りますが、ここでは、 $x$  の関数  $g$  が入ります。なお、 $g$  には  $\&$  のマークはつけません。もちろん、 $g(x)$  の区間の下限は定数である必要はなくて  $x$  の関数でかまいません。

同様にして、三重積分の例を挙げます。

プログラム

```
new; cls;
fn f(x,y,z)=x.*y.*z;
fn g1(x)=x^2+x./4+3;
fn g2(x)=3;
fn h1(x,y)=x+y;
fn h2(x,y)=0;
x1={1,0}; g=&g1 | &g2; h=&h1 | &h2;
a=intgrat3(&f,x1,g,h);
print a;
```

画面表示

10.372058

上の例は、関数  $f(x,y,z)=xyz$  に対して、二重積分までは同じで、 $z$  に相当する区間には、上限  $h1(x,y)$  には  $x+y$ 、下限  $h2(x,y)$  には  $0$  として、それらを垂直方向にマージして  $h$  としたのちに、組込み関数 `intgrat3` で三重積分したものです。 $g$  と同様、 $h$  にも `intgrad3` の関数内では  $\&$  のマークは必要ありません。

**ポイント**   `intgrat2( &関数名, x の区間, g(x) )`                       $x$  の区間を介した二重積分

**ポイント**   `intgrat3( &関数名, x の区間, g(x), h(x,y) )`    $x$  の区間を介した三重積分

このように、関数定義 `fn` を用いれば、簡単に見た目の数学と同じように 1 行関数が定義できて、微積分関係の組込み関数が扱えるようになります。もちろん、公式マニュアルその他に書いてあるように `proc` を使って関数を定義することも可能ですが、1 行の関数の定義にそこまでする必要はありません。組込み関数で呼び出すだけの 1 行で表現できる関数には、この `fn` の使用を推奨します。